

EXCEL 2003 - FONCTIONS DE CALCUL

Les **fonctions Excel** sont nombreuses et permettent de réaliser des calculs simples et complexes. Vous trouvez ainsi des fonctions pour réaliser des calculs *statistiques*, *mathématiques* ou encore *financiers*. Certaines permettent même de poser des raisonnements logiques comme la fonction **SI**. La structure des fonctions de calcul se compose de trois éléments :

- ▶ **Signe =** : Tout calcul excel commence par ce symbole,
- ▶ **Nom de la fonction** : Le nom de la fonction employé suit le signe égal comme **SOMME** ou **SI**,
- ▶ **Les paramètres** : Après avoir nommé la fonction, vous devez ouvrir les parenthèses pour fournir les paramètres du calcul (Souvent des cellules) à la fonction. Ex: **=SOMME(A1:B1)**.


QUELQUES FONCTIONS SIMPLES

Certaines fonctions sont incontournables car leur emploi est trivial et elles fournissent des résultats de synthèse forts intéressants. D'autres sont plus complexes mais nécessaires pour poser des raisonnements et aller plus loin.

Nom Fonction	Exemple	Description
SOMME	=SOMME(A1:A4)	Calcule la somme des valeurs contenues dans les cellules passées dans les parenthèses.
MOYENNE	=MOYENNE(A1:A4)	Calcule la moyenne des valeurs contenues dans les cellules passées dans les parenthèses.
MAX	=MAX(A1:A4)	Extrait et affiche la plus grande des valeurs contenues dans les cellules passées entre parenthèses.
MIN	=MIN(A1:A4)	Extrait et affiche la plus petite des valeurs contenues dans les cellules passées entre parenthèses.
NB	=NB(A1:A4)	Compte et affiche le nombre de cellules contenant une valeur.
NB.SI	=NB.SI(A1:A4;"valeur")	Compte et affiche le nombre de cellules identiques à la valeur passée en deuxième argument.
SI	=SI(A1>10;"Reçu";"Éliminé")	Permet de vérifier un critère et d'envisager l'un ou l'autre cas selon que le critère est vérifié ou non. Ainsi dans l'exemple, si la valeur de la cellule A1 est <i>supérieure</i> à 10, le résultat de la fonction affiche <i>Reçu</i> , <i>Éliminé</i> sinon.


Comme vous le remarquez, certaines fonctions utilisent du **texte** dans leurs **paramètres** soit comme critère soit comme valeur à afficher. Dans ce cas, vous devez l'encadrer de **guillemets** pour spécifier à **Excel** qu'il s'agit de **données textuelles** et non **numériques** ou de **date**.

LA SOMME AUTOMATIQUE

Avec la fonction **Somme automatique**, **Excel** facilite l'addition de valeurs qui est l'un des calculs les plus fréquents dans une **feuille de calcul Excel**. Un bouton lui est dédié sur la barre d'outils standard . Vous remarquez que ce bouton est accompagnée d'une flèche permettant d'accéder rapidement à certaines autres fonctions mais aussi à l'assistant fonction (**Autres fonctions...**).

- Créez un nouveau classeur dans Excel,
- Réalisez le tableau simple de la figure,
- Sélectionnez la première cellule de total (B7),
- Cliquez sur le bouton **Somme automatique**,

La fonction **SOMME** s'affiche dans la cellule avec entre parenthèses, une suggestion de la liste des cellules à sommer, ici **B2:B6**. Si la suggestion n'est pas la bonne, il vous suffit de cliquer et glisser la souris sur une autre plage de cellules pour changer les paramètres de la fonction.



	A	B	C	D
1	DESTINATION	JUIN	JUILLET	AOÛT
2	Paris	160	270	320
3	Venise	70	195	280
4	Londres	140	168	155
5	Barcelone	145	110	210
6	Genève	65	112	160
7	TOTAL	=SOMME(B2:B6)		
8		SOMME(nombre1; [nombre2]; ...)		

- Validez ce calcul en réalisant la combinaison de touches **CTRL + ENTREE**.

La fonction **SOMME** affiche le résultat du calcul dans la cellule où elle est écrite.


Remarque : Notez la présence des **deux points (B2:B6)** pour l'énumération des cellules à sommer dans les parenthèses de la fonction **SOMME**. Cette syntaxe est une partie du langage Excel. Les deux points signifient de **B2 à B6**. Concrètement toutes les cellules entre B2 et B6 sont sommées, ce qui simplifie l'écriture. L'autre symbole pouvant remplacer les deux points est le **point virgule (;)**. Dans ce cas, la somme automatique additionne seulement **B2 et B6**.

Remarque : Généralement, une saisie est validée par la touche **Entrée**. Mais cette action a pour effet de sélectionner la cellule du dessous. Le fait de valider une saisie par la combinaison **CTRL + Entrée** permet de conserver la cellule active. Ainsi, il est moins fastidieux de tirer la poignée dans la foulée pour reproduire un calcul.

SOMME AUTOMATIQUE ET INSTANTANÉE

Il existe une autre méthode, plus rapide que celle décrite plus haut pour réaliser une somme automatique, lorsque le tableau s'y prête. La ligne du total étant vide,

- Sélectionnez ensemble toutes les cellules à sommer ainsi que les cellules dans lesquelles doivent apparaître les résultats,
- Cliquez sur le bouton **Somme automatique**,




	A	B	C	D
1	DESTINATION	JUIN	JUILLET	AOÛT
2	Paris	160	270	320
3	Venise	70	195	280
4	Londres	140	168	155
5	Barcelone	145	110	210
6	Genève	65	112	160
7	TOTAL			

Tous les résultats s'affichent instantanément dans les trois cellules du total. Cette technique permet d'éviter l'étape intermédiaire qui consiste à valider la plage de sélection à sommer que suggère **Excel**.

INSÉRER UNE FONCTION


La boîte de dialogue **Insérer une fonction** recense toutes les fonctions de calcul, les classe par catégorie, propose un assistant pour les utiliser et offre une aide avec des exemples explicites sur l'utilisation de chacune d'elle. Lorsque vous ne connaissez pas ou peu la fonction, il est donc conseillé de faire appel à ses services.

- Créez le petit tableau de la **moyenne, Plus grand, Plus petit** ci-contre, sous le précédent tableau,



	A	B	C	D
1	DESTINATION	JUIN	JUILLET	AOÛT
2	Paris	160	270	320
3	Venise	70	195	280
4	Londres	140	168	155
5	Barcelone	145	110	210
6	Genève	65	112	160
7	TOTAL	580	855	1125
8				
9				
10	Moyenne			
11	Plus grand			
12	Plus petit			

- Sélectionnez la première cellule de la moyenne (**B10**),
- Cliquez sur la **flèche** associée au bouton **Somme automatique**,
- Dans le menu, cliquez sur **Autres fonctions...**,




La boîte de dialogue **Insérer une fonction** apparaît. Par défaut, les fonctions sont classées selon la catégorie **Les dernières utilisées**. Mais il suffit de dérouler cette liste et de choisir la catégorie adéquate. Si vous n'avez pas idée de la catégorie de la fonction mais que vous connaissez son nom, vous pouvez choisir la catégorie **Tous**. Elles seront alors triées par ordre alphabétique croissant.

- Sélectionnez la catégorie **Tous**,
- Au clavier, tapez la lettre **M**,


Ainsi vous sélectionnez directement la première des fonctions dont le nom commence par la lettre **M**.

- Utilisez la barre de défilement pour faire apparaître les fonctions situées plus bas, jusqu'à afficher **Moyenne**,
- Sélectionnez **Moyenne** dans la liste,




Notez la présence d'une brève description de la fonction sélectionnée en bas de la boîte de dialogue, juste sous la liste. De même, vous remarquez la présence d'un lien **Aide sur cette fonction** qui propose des exemples judicieux d'application de la fonction. Il est parfois plus simple de se diriger directement vers cette rubrique pour apprendre à utiliser une fonction.

- Cliquez sur le bouton **Ok** pour utiliser la fonction **Moyenne**,



L'assistant de la fonction se déclenche. Un bref descriptif explique comment remplir les arguments de la fonction. La fonction **moyenne** réalise la moyenne des valeurs des cellules sélectionnées. Les zones **Nombre1** et **Nombre2** correspondent aux plages de cellules que vous pouvez sélectionner pour les inclure dans le calcul de la moyenne. Ces plages de cellules ne peuvent être constituées que d'une seule cellule.

- Cliquez dans la zone de saisie **Nombre1** pour l'activer,
- Sélectionnez les cellules **B2 à B6** sur la feuille,



Excel écrit alors **B2:B6** signifiant que vous allez réaliser la moyenne des cellules allant de **B2** à **B6**. Vous pourriez sélectionner une autre plage de cellules dans la zone **Nombre2**, **C2** à **C6** par exemple. La fonction résultante serait **=MOYENNE(B2:B6;C2:C6)** selon la syntaxe expliquée auparavant, signifiant que vous réalisez la moyenne des cellules **B2** à **B6** et **C2** à **C6**. Dans ce cas, vous constatez l'apparition d'une nouvelle zone de saisie **Nombre3** et ainsi de suite. Dans le cas présent, nous nous contentons de la moyenne pour le mois de juin (**B2:B6**).

- Cliquez sur **Ok** pour valider le calcul.

Le résultat s'affiche dans la cellule. **L'assistant fonction** permet donc de ne pas saisir le calcul et de ne pas se soucier de la syntaxe.

- Tirez la poignée de deux cases sur la droite pour reproduire le calcul.

6	Genève	65	112	160
7	TOTAL	580	855	1125
8				
9				
10	Moyenne	=MOYENNE(B2:B6)	171	225
11	Plus grand			
12	Plus petit			
13				

LES ASSISTANTS INFO-BULLES

Lorsque vous utilisez des fonctions simples il est quasiment préférable de se laisser guider par les *info-bulles* qui s'affichent en même temps que vous saisissez la fonction plutôt que d'utiliser l'*assistant*. Les fonctions *Max* et *Min* s'utilisent exactement comme la fonction *moyenne*. Il suffit de sélectionner en argument la plage de cellule sur laquelle elles doivent extraire la valeur caractéristique. Les info-bulles expliquent comment remplir la fonction en cours.

- Sélectionner la cellule **B11**,
- Commencez la saisie de la fonction par **=MAX(**

Vous constatez l'apparition de la bulle d'aide : **MAX(nombre1; [nombre2];...)**. *nombre1* et *nombre2* pouvant correspondre à une cellule ou à une plage de cellules. Les trois petits points signifient que la liste des arguments après le premier est facultatif. Ainsi nous pouvons nous contenter de ne sélectionner qu'une cellule ou une plage de cellule.

- Sélectionnez alors la plage de cellule **B2 à B6**,
- Fermez la parenthèse,
- Validez le calcul par **CTRL + Entrée**,

Il ne vous reste plus qu'à tirer la poignée de deux cases supplémentaires sur la droite pour reproduire le calcul. La poignée est le petit carré noir situé en bas à droite de la cellule. Lorsque vous placez votre curseur sur ce carré, il se transforme en une petite croix noire, signifiant que vous pouvez glisser la souris sur d'autres cellules pour reproduire le calcul sans le ressaisir.

	A	B	C	D
1	DESTINATION	JUIN	JUILLET	AOUT
2	Paris	160	270	320
3	Venise	70	195	280
4	Londres	140	168	155
5	Barcelone	145	110	210
6	Genève	65	112	160
7	TOTAL	580	855	1125
8				
9				
10	Moyenne	116	171	225
11	Plus grand	=MAX(
12	Plus petit	MAX(nombre1; [nombre2]; ...)		

		580	855
		116	171
		160	
			+

Remarque : Si vous ne fermez pas la parenthèse de la fonction, au moment de valider, Excel le fera pour vous.

PLUTÔT QUE DE TIRER LA POIGNÉE...

Dans tous les calculs que nous avons vus jusqu' alors, nous avons d'abord posé la formule sur une cellule pour reproduire ensuite la logique sur les autres cellules en tirant la poignée. Mais une technique méconnue permet de valider en calcul tout en le reproduisant simultanément sur les cellules désirées. Elle consiste à premièrement sélectionner toutes les cellules où doivent apparaître le calcul. Ensuite, il faut réaliser le calcul comme s'il s'agissait de la première cellule sélectionnée. Enfin, il faut valider par **CTRL + Entrée**. Voyons cela avec le calcul du minimum.

- Sélectionnez les cellules **B12 à D12**,

Par défaut dans une plage sélectionnée, c'est la première cellule qui est active. C'est ainsi que nous débutons le calcul.

- Saisissez la formule **=MIN(B2:B6)**,

Il est préférable de ne pas saisir la plage de cellules mais de la sélectionner.

- Validez ce calcul en enfonçant ensemble les touches **CTRL et Entrée**.

Vous constatez que le résultat du calcul est répercuté dans toutes les cellules présélectionnées.



Remarque : La casse des noms de fonction n'a pas d'importance. Aussi pouvez vous saisir une fonction aussi bien en majuscules qu'en minuscules. ex : =min(B2:B6) ou =MIN(B2:B6).


ACCÈS RAPIDE AUX FONCTIONS COURANTES

Les fonctions les plus simples et les plus courantes bénéficient d'un accès rapide, dérivé de l'assistant fonction.

- Sélectionnez la cellule dans laquelle vous souhaitez réaliser le calcul,
- Cliquez sur la flèche du bouton **Somme automatique**,

Vous pouvez choisir entre **Somme**, **moyenne**, **Compteur**, **Max** et **Min**.

- Cliquez sur la fonction de votre choix,
- Et laissez vous guider par l'**info-bulle**.



LES OPÉRATEURS DE CALCUL


Bien sûr il existe les fonctions Excel pour réaliser des calculs simples ou complexes, mais l'utilisation des opérateurs n'en demeurent pas moins vrai lorsqu'il s'agit de réaliser une addition, soustraction, division, multiplication ou encore un calcul de puissance. Le tableau ci-dessous liste ces opérateurs.

Opérateur	Opération	Exemple
+	Addition	A1+A2
-	Soustraction	A1-A2
*	Multiplication	A1*A2
/	Division	A1/A2
%	Pourcentage	55%
^	Exposant	A1^A2

La figure ci-contre illustre le calcul du nombre 5 élevé à la **puissance 3**.

- Réalisez ce calcul en saisissant des valeurs arbitraires dans des cellules,

Remarque, calculs dynamiques : La philosophie des calculs dans Excel consiste à ce que toutes les opérations soient bâties sur des **variables** comme c'est le cas ici et non sur des constantes. En effet, nous n'élevons pas le nombre 5 qui est une constante à la puissance 3 qui est une autre constante, mais le nombre contenu dans la cellule **B20** élevé au nombre contenu dans la cellule **C20**. Comme les calculs sont dynamiquement liés aux cellules sur lesquels ils sont bâtis, si nous changeons les valeurs de l'une ou l'autre cellule, le résultat change instantanément car il est recalculé.



- En **B20**, saisir **8** à la place de **5**,
- Validez ce calcul.

Vous constatez en effet que le résultat n'est plus **125** mais **512**.

PRINCIPE DE RAISONNEMENT - FONCTION SI

La fonction **SI** permet de poser un critère sur une cellule. Si ce critère est vérifié, elle permet de prendre une direction en affichant un résultat par exemple, s'il n'est pas vérifié, elle permet de prendre une autre direction en affichant un autre résultat.

La construction de la fonction **SI** n'est donc pas aussi simple que les fonctions que nous avons vues jusqu'alors. Il ne suffit pas de sélectionner une plage de valeurs après avoir ouvert la parenthèse.

La fonction **SI** requiert trois paramètres :

- ▶ *Tout d'abord, le critère à vérifier. Il s'agit souvent d'une cellule comparée à une valeur;*
- ▶ *Ensuite et en deuxième argument, il faut indiquer quoi faire lorsque le critère est vrai. Il peut s'agir d'un texte ou d'écrire dans la cellule du calcul mais aussi d'un calcul à réaliser;*
- ▶ *Enfin, en troisième et dernier argument, il faut lui indiquer quoi faire lorsque le critère est faux. Là aussi, il peut s'agir d'un texte comme d'un calcul.*

	A	B	C
1	Conseil de classe		
2			
3	Elèves	Moyenne générale	Avis du conseil
4	Marc	12	
5	Sophie	18	
6	Fred	15	
7	Anne	9,6	
8	Christine	14	
9	Luc	12	
10	Roger	9,9	
11	Tifanie	13	
12	Paul	8,4	
13			

En résumé : **=SI(CRITERE; QUOI FAIRE ALORS; QUOI FAIRE SINON).**

La fonction **SI** comptant nécessairement *trois paramètres* compte forcément *deux points virgules* pour séparer ces paramètres.

- *Sur une feuille vierge, réalisez le tableau présenté par la figure ci-dessus,*

Le principe est simple. Des moyennes générales correspondent à des élèves. Dans la colonne vide *Avis du conseil*, il s'agit d'inscrire dynamiquement si l'élève *passé* (Moyenne supérieure ou égale à 10) ou s'il *redouble*. Le critère consiste donc à comparer la valeur de chaque moyenne générale à la valeur 10.

- *Sélectionnez la première cellule de la colonne Avis du conseil (C4),*
- *Saisissez la fonction suivante : =SI(B4>=10;"Passe";"Redouble").*

Sa traduction est simple. Si la moyenne est *supérieure ou égale à 10* alors la fonction écrit le texte *Passe* sinon elle écrit le texte *Redouble*.

Remarque : **>=** signifie **supérieur ou égal**. **<=** signifie **inférieur ou égal**. **<>** signifie **différent de**.

- *Validez ce calcul par CTRL + Entrée,*
- *Tirez la poignée sur toute la colonne Avis du conseil.*

Vous constatez que tous les résultats sont instantanément reproduits en s'adaptant à chaque moyenne du tableau.

	B	C	D
	Conseil de classe		
	Moyenne générale	Avis du conseil	
	12	=SI(B4>=10;"Passe";"Redouble")	
	18		
	15		
	9,6		
	14		
	12		

Bien sûr il demeure un inconvénient. La fonction est en effet bâtie sur des *constantes* comme le sont la valeur **10** mais aussi les textes *Passe* et *Redouble*. Nous verrons comment palier le problème avec les *références absolues*. De même, la fonction ne permettant de poser qu'un seul *critère* ne permet d'envisager que *deux cas*. Voyons dès maintenant comment il est possible d'augmenter le nombre de critères avec la technique **d'imbrication**.

IMBRICATIONS DE FONCTIONS SI

Voici la problématique. Lorsque la moyenne de l'élève est *supérieure ou égale à 10*, l'élève *pass*e. Cela ne change pas. Lorsque sa moyenne est *inférieure à 9,5*, il *redouble*. Enfin, lorsque la moyenne est *comprise entre 9,5 et 10*, son *cas est à étudier*.

Il y a trois cas à envisager : *Pass*e, *Redouble* et *à étudier*. Un seul critère ne suffit plus car il peut être soit vrai soit faux. Nous devons ajouter un critère en imbriquant une seconde fonction **SI** dans la première au niveau du dernier argument (Celui qui exprime le Sinon).

Traduction concrète : Si la *moyenne* est ≥ 10 , la fonction écrit *Pass*e sinon si elle est $\geq 9,5$, la fonction écrit *à étudier*, sinon elle écrit *redouble*. Il y a donc deux critères pour trois cas.


Vous avez compris la subtilité, l'expression *Compris entre* n'a pas de traduction en langage *Excel*. Tout dépend de *l'ordre des critères*. Lorsque vous exprimez, *si elle est $\geq 9,5$* , signifie que le premier critère (≥ 10) n'est pas vrai, donc vous êtes inférieur à 10. Si en même temps, vous vérifiez ce deuxième critère, vous déduisez que vous êtes dans *l'intervalle 9,5 à 10*.

- **Supprimez** tous les résultats de la colonne *Avis du conseil*,
- **Sélectionnez** toutes les cellules de cette colonne,
- **Saisissez** la formule : `=si(B4>=10;"Passe";si(B4>=9,5;"A étudiant";"Redouble"))`,

Bien sûr il faut fermer deux parenthèses puisque vous en ouvrez deux, étant donnée que la seconde fonction **SI** est imbriquée dans la première.

- **Validez** le calcul par **CTRL + Entrée**.

Vous le répercutez ainsi sur toutes les cellules de la sélection. Bien sûr si vous souhaitez encore augmenter le nombre de critères, il vous suffit d'imbriquer une troisième fonction **SI** dans la deuxième selon le même principe et ainsi de suite...



Conseil de classe					
Moyenne générale	Avis du conseil				
12	=si(B4>=10;"Passe";si(B4>=9,5;"A étudiant";"Redouble"))				
18					
15					
9,6					
14					
12					

ET, OU...COMMENT VÉRIFIER PLUSIEURS CONDITIONS ENSEMBLES

Les fonctions **ET** et **OU** s'utilisent de la même façon et possèdent la même syntaxe. Il suffit, dans les parenthèses de la fonction, de lister tous les *critères* à vérifier en les séparant de *points virgules*. La fonction **ET** exige que *tous les critères* listés soient *vrais*. Si l'un d'entre eux est faux, tout le test est faux. La fonction **OU** en revanche exige que seulement l'un des critères listés soit vrai. Si plusieurs critères ou tous sont vérifiés, le test est toujours vrai. Pour que le test échoue, il faut que tous les critères soient faux. Voyons la syntaxe :

ET(Critère1; Critère2, Critère3; ...)
OU(Critère1; Critère2, Critère3; ...)

DreamLive : www.dreamlive.fr

Migration Excel 2003

Fonctions de calcul



Voici la problématique proposée pour la mise en pratique de ces fonctions. L'établissement Evarist exige que l'étudiant obtienne une moyenne supérieure à 10 en Français **et** en Math pour passer en Terminale S. Si l'une des deux moyennes est inférieure à 10, alors l'étudiant doit redoubler.

	A	B	C	D	E
1	Etablissement Evarist : Paul			Etablissement Evarist : Jean	
2	Français	12		Français	13
3	Anglais	15		Anglais	16
4	Mathématiques	11		Mathématiques	8
5					
6	Moyenne	12,67		Moyenne	12,33
7					
8	Délibération Terminale S		Passage	Délibération Terminale S	
9					
10					

Les deux conditions devant être vérifiées ensemble et non l'une après l'autre, vous devez cette fois imbriquer une fonction **ET** dans une fonction **SI** au niveau de l'énumération du critère. En conservant la logique de la fonction **SI**, si les deux critères sont vérifiés, **alors** il passe **sinon** il redouble. Deux cas sont proposés, Paul et Jean. Le premier vérifie les deux conditions tandis que le second n'en vérifie qu'une.

	A	B	C	D	E
1	Etablissement Evarist : Paul			Etablissement Evarist : Jean	
2	Français	12		Français	
3	Anglais	15		Anglais	
4	Mathématiques	11		Mathématiques	
5					
6	Moyenne	12,67		Moyenne	
7					
8	Délibération Terminale S		=SI(ET(B2>10;B4>10);"Passage";"Redoublement")		
9					
10					

- Sélectionnez la cellule **B8**,
- Saisissez la formule ci-dessous :
=SI(ET(B2>10;B4>10);"Passage";"Redoublement")
- Validez le calcul par **CTRL + Entrée**.

La fonction affiche le résultat **Passage** puisque les deux critères sont vérifiés (**ET(B2>10;B4>10)**).

- Reproduisez cette formule pour l'autre étudiant, **Jean** :
=SI(ET(E2>10;E4>10);"Passage";"Redoublement")
- Validez par **CTRL + Entrée**.

Cette fois le résultat affiche **Redoublement** puisque seulement l'une des deux conditions est vraie.

- Remplacez la fonction **ET** par la fonction **OU** dans les deux calculs.

Maintenant les deux résultats affiche **Passage** puisque dans les deux cas, au moins une des conditions est vérifiée.

	A	B	C	D	E
1	Etablissement Evarist : Paul			Etablissement Evarist : J	
2	Français	12		Français	
3	Anglais	15		Anglais	
4	Mathématiques	11		Mathématiques	
5					
6	Moyenne	12,67		Moyenne	
7					
8	Délibération Terminale S		=SI(OU(B2>10;B4>10);"Passage";"Redoublement")		
9					
10					

Remarque, reprendre la saisie d'un calcul : Plusieurs techniques permettent de reprendre la saisie d'un calcul déjà validé. Vous pouvez par exemple modifier le calcul depuis la **barre de formule** en ayant préalablement sélectionné la cellule. Vous pouvez aussi **double cliquer** sur la cellule. La formule s'affiche et vous pouvez ainsi la modifier directement sur la feuille. La touche **F2** du clavier permet elle aussi de rendre active la saisie d'une cellule qui contient déjà des informations.

	A	B	C	D
1	Etablissement Evarist : Paul			Etablis
2	Français	12		Français
3	Anglais	15		Anglais
4	Mathématiques	11		Mathématique

Le saviez-vous ? Le fait d'enfoncer la touche **Suppr** du clavier ne supprime que le contenu d'une cellule (Textes, formules...) et non sa mise en forme (Couleurs, bordures...).



LA CONCATÉNATION

La **concaténation** est une opération qui consiste à **assembler les chaînes de caractères**, à les mettre bout à bout. Elle permet d'assembler des textes, mais aussi et pourquoi pas, un texte à un résultat de calcul pour rendre plus explicite la valeur de synthèse fournie, par exemple. L'opérateur de concaténation dans Excel est le **ET COMMERCIAL** (ou *esperluette*, touche 1 du clavier), soit **&**.

- Commencez par réaliser le petit tableau de la figure,
- Sélectionnez la première cellule de la colonne **Nom complet** soit **D7**,
- Saisissez la formule suivante :
=C7 & B7

	A	B	C	D
1				
2				
3				
4				
5				
6				
	Nom	Prénom	Nom complet	
7	Rouana	Marie		
8	Ogne	Paul		
9	Gathor	Nathalie		
10	Douche	Urbain		
11				
12				
13				
Total				
15				
16				

Nous concaténons ainsi le prénom et le nom.

- Validez ce calcul par **CTRL + Entrée**.

Le résultat obtenu est bien celui du prénom suivi du nom dans la même cellule. Cependant, il manque un espace pour les séparer. Concrètement nous devons **concaténer** le prénom avec un espace que nous devons **concaténer** avec le nom. *L'espace doit être écrit entre guillemets dans une formule.*

- Sélectionnez la cellule du résultat (**D7**),
- Enfoncez **F2** pour activer la saisie,
- Modifiez la formule comme suit :
=C7 & " " & B7
- Validez le calcul par **CTRL + Entrée**,
- Reproduisez la formule en **tirant la poignée**.

Le résultat est parfait puisque cette fois le prénom et le nom sont séparés d'un espace.

Nom	Prénom	Nom complet
Rouana	Marie	C7 & B7
Ogne	Paul	
Gathor	Nathalie	
Douche	Urbain	

Nom	Prénom	Nom complet
Rouana	Marie	C7 & " " & B7
Ogne	Paul	
Gathor	Nathalie	
Douche	Urbain	

Nom	Prénom	Nom complet
Rouana	Marie	Marie Rouana
Ogne	Paul	Paul Ogne
Gathor	Nathalie	Nathalie Gathor
Douche	Urbain	Urbain Douche

RÉSULTAT STATISTIQUE ET CONCATÉNATION

La fonction **NB.SI** est particulièrement intéressante car elle permet de compter, sur une plage de cellules données, le nombre de cellules vérifiant le critère défini. Elle requiert donc deux paramètres : La **plage de cellule** sur laquelle dénombrée et le **critère**. Elle fournit un résultat numérique brute qu'il est souvent plus facile d'interpréter en le concaténant avec un texte d'explication.

- Récupérer le tableau que vous avez réalisé sur les fonctions **SI imbriquées**,

Ou bien,

- Refaites le.

	A	B	C
1	Conseil de classe		
2			
	Elèves	Moyenne générale	Avis du conseil
4	Marc	12	Passe
5	Sophie	18	Passe
6	Fred	15	Passe
7	Anne	9,6	A étudier
8	Christine	14	Passe
9	Luc	12	Passe
10	Roger	9,9	A étudier
11	Tifanie	13	Passe
12	Paul	8,4	Redouble
13			
14			
Total			
16			

Objectif : Afficher dans la cellule du total, le nombre d'admissions. Vous devez donc compter, à l'aide de la fonction **NB.SI**, le **nombre de notes supérieures à 10** dans la colonne **Moyenne générale**.



- Sélectionnez la cellule du total (**B15**),
- Saisissez la formule suivante :
=nb.si(B4:B12;">10")

Le premier argument est donc la plage de cellules sur laquelle vous devez compter les cellules correspondant au critère. Le second argument est donc le critère lui-même. Notez que ce dernier est placé entre *guillemets*.

- Validez par **CTRL + Entrée**.

Le résultat affiche 6. Certes il y a 6 admissions mais ce résultat n'est pas très évocateur pour l'utilisateur non averti. Il convient donc de le concaténer avec le texte *admissions* par exemple.

- Sélectionnez de nouveau la cellule du calcul (**B15**),
- Enfoncez **F2** pour activer sa saisie,
- Puis, modifiez la formule comme suit :
=NB.SI(B4:B12;">10") & " admission(s)"

N'oubliez pas l'espace avant le texte *admission(s)* dans les guillemets, afin de séparer le texte du résultat numérique.

- Validez par **CTRL + Entrée**.

Cette fois le résultat est explicite pour quiconque puisqu'il affiche **6 admission(s)**. Bien sûr le texte est invariant. Le résultat numérique lui varie en fonction du tableau puisqu'il est issu d'un calcul.

- Modifiez la valeur **9,6** de Anne en **12**,
- Puis validez.

En effet, le résultat devient instantanément **7 admission(s)**.

	A	B	C
1	Conseil de classe		
2			
3	Elèves	Moyenne générale	Avis du conseil
4	Marc	12	Passe
5	Sophie	18	Passe
6	Fred	15	Passe
7	Anne	9,6	A étudier
8	Christine	14	Passe
9	Luc	12	Passe
10	Roger	9,9	A étudier
11	Tifanie	13	Passe
12	Paul	8,4	Redouble
13			
14			
15	Total	=nb.si(B4:B12;">10")	
16			

9	Luc	12	Passe
10	Roger	9,9	A étudier
11	Tifanie	13	Passe
12	Paul	8,4	Redouble
13			
14			
15		=NB.SI(B4:B12;">10") & " admission(s)"	
16			
17			

	Elèves	Moyenne générale	Avis du conseil
4	Marc	12	Passe
5	Sophie	18	Passe
6	Fred	15	Passe
7	Anne	12	Passe
8	Christine	14	Passe
9	Luc	12	Passe
10	Roger	9,9	A étudier
11	Tifanie	13	Passe
12	Paul	8,4	Redouble
13			
14			
15	Total	7 admission(s)	
16			

LA RECHERCHE D'INFORMATIONS

Excel propose notamment deux fonctions très puissantes pour extraire de l'information de bases de données selon critère. Il s'agit des fonctions **RECHERCHEV** et **RECHERCHEH**. La première permet de *rechercher verticalement* tandis que la seconde permet de *rechercher horizontalement*. RechercheV est plus souvent utilisée dans la mesure où la majorité des tableaux sont présentés sous forme de colonnes et non de lignes.


- Créez un nouveau classeur,
- Nommez la feuille **Base**,
- Réalisez le tableau de la figure.

Ce tableau sert de *base de données* duquel vous allez extraire de l'information sur une autre feuille à l'aide de la fonction **RechercheV**.

	A	B	C	D
1	NOM	PRENOM	C.POSTAL	VILLE
2	DECAJOU	Benoît	26000	VALENCE
3	GATHOR	Nathalie	26300	BOURG DE P
4	DOUCH	Urbain	26100	ROMANS
5	OPILON	Marthe	07000	PRIVAS
6	HOCHET	Eric	7800	LA VOULTE
7	MUDAT	Albert	26270	LORIOLE
8	AUDINE	Anne	26300	BOURG DE P
9	HOCHON	Paul	26100	ROMANS
10	DARMES	Jean	07000	PRIVAS
11	GALLES	Martin	26000	VALENCE
12	CHANDESSE	Irène	73000	CHAMBERY
13	ADRENE	Aline	26130	ST PAUL
14	MEYABALPA	Jésus	26250	PIERRELATTE
15	YGINAL	Laure	26000	VALENCE
16	CASSELANE	Etienne	26200	MONTELMAR
17				



Remarque, préfixe zéro pour les cellules numériques : Dans le cas des cellules numériques, lorsque vous débutez la saisie par un zéro, vous constatez qu'Excel le masque. En effet, le tableur Excel manipulant des nombres estime l'information inutile. Mais dans le cas des codes postaux, ce comportement s'avère gênant. Ainsi pour un code postal comme 07500, débutez la saisie par une apostrophe (touche 4 du clavier), '07500. Vous conservez ainsi le zéro en préfixe et l'apostrophe n'apparaît pas dans la cellule. D'ailleurs, par le biais d'une balise active, Excel vous informe que la donnée numérique est dorénavant stockée sous forme de texte.



- Sur une nouvelle feuille du même classeur, réalisez la petite fiche illustrée ci-dessous,
- Nommez cette feuille, **fiche**.

Principe : Vous saisissez un nom issue de la base de donnée en cellule C3. Lorsque vous validez, vous obtenez instantanément les informations correspondant à ce nom, comme le prénom et la ville grâce à la fonction **RechercheV**.

	A	B	C
1			
2			
3			
4		Nom :	
5			
6		Prénom :	
7			
8		Ville :	
9			
10			

La fonction **RechercheV** requiert quatre paramètres. Tout d'abord il faut lui fournir l'élément à rechercher pour extraire l'information. Ici, il s'agit du nom dont la recherche fournira le prénom et la ville. Appelons de paramètre **el_recherche**. Ensuite, il faut lui fournir la plage de cellule correspondant à la base de données contenant ces informations. Ici, il s'agit du **tableau de la feuille Base**. Appelons ce paramètre **tableau**. Ensuite, il faut lui indiquer en numérique, le numéro de colonne où se trouve l'information à extraire. Ici le prénom est en colonne 2 et la ville en colonne 4. Appelons ce paramètre **num_colonne**. Enfin, le dernier paramètre indique si la fonction doit tenter de se rapprocher du résultat lorsqu'elle ne trouve pas. Il admet deux valeurs, **VRAI** ou **FAUX**. Appelons ce paramètre, **test**.

En résumé, la fonction **RechercheV** s'écrit ainsi :
=RechercheV(el_recherche, tableau, num_colonne, test)

- Sélectionnez la cellule du prénom (C6),
- Commencez la saisie de la formule comme suit :
=recherchev(C4;



Vous désignez ainsi la cellule du nom (à sélectionner) comme élément de recherche. Après le point virgule, vient le tableau dans lequel doit s'effectuer la recherche.

- Durant la saisie de la formule, activer la feuille **Base**,
- Puis, sélectionnez le tableau entier, lignes d'en-têtes comprises :
=recherchev(C4;base!A1:D16

Le tableau ainsi sélectionné s'écrit sous forme de plage de cellule précédé du nom de la feuille (Base!) sur laquelle il se trouve.

- Ajoutez un deuxième point virgule pour passer au troisième argument,
- Saisissez le chiffre **2** suivi d'un point virgule :
=recherchev(C4;base!A1:D16;2;

En effet, vous cherchez à extraire le prénom qui se trouve dans la colonne 2 du tableau.

- Enfin terminez la saisie de la formule en saisissant le texte **faux** et en fermant la parenthèse,
- Validez cette formule par **CTRL + Entrée**.
=recherchev(C4;base!A1:D16;2;faux)



Vous terminez par la valeur Faux car la fonction ne doit pas tenter de se rapprocher du résultat si elle ne trouve pas. Le résultat obtenu est une erreur de calcul #N/A ! Bien sûr, comme la cellule du nom est vide, vous ne fournissez aucun élément de recherche à la fonction, par conséquent le résultat retourné est indisponible.

- Sélectionnez la cellule du nom (C4),
- Saisissez un nom de la base, par exemple **Douch**,
- Validez la saisie.

	A	B	C	D
1				
2				
3				
4		Nom :	douch	
5				
6		Prénom :	Urbain	
7				
8		Ville :		
9				

Instantanément, l'information correspondante (Le prénom) est retournée par la fonction RechercheV et s'affiche en C6. Bien entendu, si vous changez de nom, le prénom correspondant est répercuté dans la cellule du prénom.

- Reproduisez cette formule pour récupérer la ville.

	A	B	C	D	E
1					
2					
3					
4		Nom :	douch		
5					
6		Prénom :	Urbain		
7					
8		Ville :	=RECHERCHEV(C4;base!A1:D16;4,FAUX)		
9					

Désormais, le fait de changer le **nom** adapte automatiquement les **Prénom** et **Ville** correspondant, recherchés dans la base de données.

Néanmoins à ce stade, comme vous l'avez constaté précédemment, si vous supprimez le contenu de la cellule du nom, vous obtenez deux messages d'erreur disgracieux en lieu et place du prénom et de la ville. Bien qu'aucune information ne soit présente dans la cellule du nom, la fonction **RechercheV** tente d'effectuer la recherche. Comme elle n'y parvient pas, elle retourne **#N/A**.

La fonction **SI** est un excellent moyen de contourner le problème pour indiquer que si la cellule est vide, aucune recherche ne doit être effectuée. Dans le cas contraire, la recherche doit être réalisée. Il s'agit donc **d'imbriquer** la fonction **RechercheV** dans la fonction **SI** comme suit :

=SI(C4="";"";RECHERCHEV(C4;base!A1:D16;2;FAUX))

Traduction : Si la cellule du nom est vide (**SI(C4="";**), alors il ne faut rien faire (**""**), sinon la fonction RechercheV doit rechercher le nom pour retourner le prénom (**RECHERCHEV(C4;base!A1:D16;2;FAUX)**). L'astuce **""** permet de désigner une cellule vide dans le cas du test mais aussi dans le cas de la saisie.

Dès lors, si le nom est absent, la formule, intelligente, ne réalise aucune recherche et ne produit aucune erreur.

	A	B	C	D
1				
2				
3				
4		Nom :		
5				
6		Prénom :		
7				
8		Ville :		
9				

Remarque, comment afficher les formules dans les cellules à la place des résultats ?

Il suffit de dérouler le menu **Outils** et de cliquer sur **Options**. Dans la boîte de dialogue **Options**, vous devez sélectionner l'onglet **Affichage** et cocher la case **Formules**. Bien sûr pour ne plus voir les formules et retrouver les résultats de calculs, il suffit de décocher cette case.

	B	C
1		
2		
3		
4	Nom :	option
5		
6	Prénom :	=SI(C4="";"";RECHERCHEV(C4;base!A1:D16;2;FAUX))
7		
8	Ville :	=SI(C4="";"";RECHERCHEV(C4;base!A1:D16;4;FAUX))
9		



RÉFÉRENCES RELATIVES - RÉFÉRENCES ABSOLUES

Vous allez étudier le principe de ces références au travers d'un cas concret.

- Créez un nouveau classeur *Excel*,
- Réalisez le tableau proposé par la figure.

	A	B	C	D	E
1				TVA	19,60%
2					
3	Articles	Prix U	Qtés	Montant HT	Montant TTC
4	ar001	16,00 €	18		
5	ar002	53,00 €	16		
6	ar003	125,00 €	9		
7	ar004	82,00 €	14		
8	ar005	154,00 €	11		

Il s'agit d'articles dont les *prix unitaires* sont référencés dans la deuxième colonne. Une certaine quantité de chacun de ces articles a été vendue comme l'exprime la colonne *Qtés*. Il s'agit maintenant de calculer le montant hors taxe (*MontantHT*) ainsi que le montant TTC (*MontantTTC*) en considérant la *TVA* à 19,6%.

Le calcul du *montant hors taxe* est une simple **multiplication** du *prix unitaire* par la *quantité* vendue.

- Sélectionnez les cellules de la colonne *MontantHT* (D4 à D8),
- Réalisez le calcul : **=B4*C4**,
- Validez par **CTRL + Entrée**.

Tous les montants hors taxe sont instantanément reproduits dans la colonne. Sans le savoir vous venez d'exploiter les **références relatives**, comme vous le faites intuitivement depuis le début. C'est le mode de fonctionnement d'Excel. Vous posez un calcul sur la ligne 4 (en **D4**) en multipliant deux cellules de cette même ligne. Lorsque vous validez par **CTRL + Entrée** ou que vous **tirez la poignée**, vous demandez à Excel de reproduire le calcul sur les lignes du dessous (5, 6, 7 et 8). Sur la ligne 5 par exemple, *Excel* ne multiplie plus les cellules de la ligne 4 mais les cellules de la ligne 5 (**B5*C5**). C'est pourquoi on dit que les **références** du calcul sont **relatives**. Si le calcul bouge, change de ligne, alors les références changent elles aussi en suivant la même logique, une ligne plus bas et ainsi de suite... Cette remarque est vraie si vous changez de colonne.

	A	B	C	D	E
1				TVA	0,196
2					
3	Articles	Prix U	Qtés	Montant HT	Montant TTC
4	ar001	16	18	=B4*C4	
5	ar002	53	16	=B5*C5	
6	ar003	125	9	=B6*C6	
7	ar004	82	14	=B7*C7	
8	ar005	154	11	=B8*C8	

Dans la majorité des cas, les **références relatives** sont adaptées au contexte du calcul comme dans cet exemple. Mais dans d'autres circonstances, comme nous allons le voir avec le calcul du *TTC*, ce comportement d'*Excel* est gênant et vous devez palier le problème pour ne pas avoir à reproduire maintes fois le calcul.

- Sélectionnez les cellules de la colonne *MontantTTC* (E4 à E8),
- Réalisez le calcul : **=D4*(1+E1)**,
- Validez par **CTRL + Entrée**.

Le calcul du *TTC* consiste en effet à ajouter les charges (*HT*TVA*) au montant hors taxe de départ (*HT*) => **HT*(1 + TVA)**. Le calcul produit un résultat cohérent dans la mesure où le *TTC* est bien supérieur au *HT*.

- Tirez la poignée du calcul sur toute la colonne.

	A	B	C	D	E
1				TVA	19,60%
2					
3	Articles	Prix U	Qtés	Montant HT	Montant TTC
4	ar001	16,00 €	18	288,00 €	344,45 €
5	ar002	53,00 €	16	848,00 €	848,00 €
6	ar003	125,00 €	9	1 125,00 €	#VALEUR!
7	ar004	82,00 €	14	1 148,00 €	396 574,30 €
8	ar005	154,00 €	11	1 694,00 €	1 438 206,00 €

Cette fois, comme vous le constatez, rien ne va plus. Certains *TTC* sont identiques au *HT*, d'autres sont exorbitants et d'autres encore conduisent à un message d'erreur (**#VALEUR!**). La raison est due à ces fameuses références relatives qui n'ont pas lieu d'être ici. En effet le premier calcul met en jeu le *HT* (D4) et la *TVA* (E1). Lorsque vous tirez la poignée une ligne plus bas, Excel choisit donc les cellules une ligne plus, soit **D5** pour le nouveau *montant HT* ce qui est logique, et **E2** pour la *TVA*. Or la cellule **E2** est *vide*. En effet, le *taux de TVA* est inscrit dans une *cellule unique* et le calcul bien qu'incrémenté sur les autres lignes doit toujours faire référence à cette même *cellule E1*. Pour palier le problème nous devons donc exploiter les **références absolues**.

- Effacez les résultats de la colonne **MontantTTC**,
- Sélectionnez de nouveau toutes les cellules (**E4 à E8**),
- Commencez à saisir le calcul comme précédemment sans fermer la parenthèse : **=D4*(1+E1**

Vous venez de sélectionner le **taux de TVA (E1)**. Il s'agit maintenant d'indiquer à Excel que cette référence est absolue, qu'elle ne doit pas changer lorsque vous réperceutez le calcul sur les cellules du dessous. Pour ce faire,

- Enfoncez la touche **F4** du clavier,

Vous constatez que deux **\$** encadrent désormais la référence à **E1 (\$E\$1)** signifiant que cette cellule est **figée** dans le calcul (**absolue**).

- Fermez la parenthèse,
- Validez par **CTRL + Entrée**.

U	Qtés	Montant HT	Montant TTC
€	18	288,00 €	=D4*(1+\$E\$1)
€	16	848,00 €	848,00 €
€	9	1 125,00 €	#VALEUR!
€	14	1 148 00 €	396 574 30 €

Cette fois les résultats sont cohérents dans la mesure où le taux de TVA est exploité dans chacun des calculs.

	A	B	C	D	E
1				TVA	0,196
2					
3	Articles	Prix U	Qtés	Montant HT	Montant TTC
4	ar001	16	18	=B4*C4	=D4*(1+\$E\$1)
5	ar002	53	16	=B5*C5	=D5*(1+\$E\$1)
6	ar003	125	9	=B6*C6	=D6*(1+\$E\$1)
7	ar004	82	14	=B7*C7	=D7*(1+\$E\$1)
8	ar005	154	11	=B8*C8	=D8*(1+\$E\$1)
9					

Remarque, deux dollars (\$) pour deux degrés de libertés : Lorsque vous enfoncez la touche **F4** du clavier, deux dollars encadrent la référence de cellule, un dollar devant le **E** et un dollar devant le **1** (**\$E\$1**). Le dollar devant le **E** bloque la cellule en colonne dans le calcul, si vous tirez la poignée à droite par exemple. Le dollar devant le **1** bloque la cellule en ligne dans le calcul, si vous tirez la poignée vers le bas comme c'est le cas ici. Dans cet exemple, nous aurions pu nous contenter de figer seulement la ligne donc. Si vous enfoncez de nouveau la touche **F4** (en saisie de formule), vous constatez qu'il ne subsiste plus qu'un dollar devant la référence de ligne (**\$E1**). Si vous continuez d'enfoncer la touche **F4**, le dollar se positionne devant la référence de colonne (**\$E1**). Et si vous enfoncez une dernière fois la touche **F4**, les dollars disparaissent rendant les deux degrés de liberté à la cellule (ligne et colonne).

EN VRAC

RÉFÉRENCE CIRCULAIRE

Nous venons de voir les **références relatives** et **absolues**. Mais que sont donc les **références circulaires**. Les **références circulaires** ne constituent pas une technique de calcul mais résultent d'une **erreur** lorsque vous incluez la cellule sur laquelle est posée la formule dans le calcul. Elle ne peut livrer aucun résultat puisqu'elle attend sa propre valeur pour calculer. Il s'agit d'une fonction qui se *mord la queue*. On parle alors de **référence circulaire**.

	A	B	C	D	E
1				TVA	20%
2					
3	Articles	Prix U	Qtés	Montant HT	Montant TTC
4	ar001	16,00 €	18	288,00 €	344,45 €
5	ar002	53,00 €	16	848,00 €	1 014,21 €
6	ar003	125,00 €	9	1 125,00 €	1 345,50 €
7	ar004	82,00 €	14	1 148,00 €	1 373,01 €
8	ar005	154,00 €	11	1 694,00 €	2 026,02 €
9				=MOYENNE(D4:D9)	
10					
11					

Dans l'exemple de la figure, la moyenne inclut la cellule où le calcul est posé (**D9**). C'est pourquoi aucun résultat n'est fourni. Dès validation, une **aide** ainsi que la barre d'outils **Référence circulaire** s'affichent. Pour palier le problème, il suffit de corriger la formule de la moyenne en excluant la cellule **D9** du calcul.


Vous pouvez cliquer sur le bouton **Repérer les antécédents** de la barre d'outils **Référence circulaire** de manière à afficher les cellules incluses dans le calcul et à isoler celle qui est incriminée.



LA FONCTION ROMAIN

Cette fonction est anecdotique. Cependant, à titre d'astuce, il est intéressant de l'utiliser pour constater qu'elle permet de transcrire un *nombre arabe* en *nombre romain*. Cette fonction **ROMAIN** n'attend qu'un seul paramètre, la cellule du nombre à transcrire.

- Créez un nouveau classeur **Excel**,
- Saisissez quelques nombres de **A1 à A6** par exemple,
- Sélectionnez les cellules de **B1 à B6**,
- Puis, saisissez la formule **=ROMAIN(A1)**,
- Validez par **CTRL + Entrée** pour reproduire le calcul.




	A	B
1	18	XVIII
2	59	LIX
3	104	CIV
4	360	CCCLX
5	680	DCLXXX
6	1220	MCCXX

Vous obtenez ainsi toutes les transcriptions romaines des nombres saisis.

LA FONCTION ESTNUM

La fonction **ESTNUM** permet de réaliser un *test logique*. Elle n'attend qu'un paramètre, la cellule à tester. Elle renvoie **VRAI** si son contenu est numérique et **FAUX** dans le cas contraire. Utilisée conjointement avec une fonction **SI** pour réaliser un critère, elle s'avère fort utile.

- Saisissez quelques nombres et textes de **A1 à A6** par exemple,
- Sélectionnez les cellules de **B1 à B6**,
- Puis, saisissez la formule **=ESTNUM(A1)**,
- Validez par **CTRL + Entrée** pour reproduire le calcul.




	A	B
1	18	VRAI
2	59	VRAI
3	texte	FAUX
4	360	VRAI
5	x220	FAUX
6	1220	VRAI

Le résultat est illustré par la figure.

LA FONCTION ESTVIDE

La fonction **ESTVIDE** est elle aussi une *fonction logique*. De la même manière, elle n'attend qu'un paramètre, la cellule à tester. Elle renvoie **VRAI** si son contenu est *vide* et **FAUX** dans le cas contraire. Son utilisation conjointe avec la fonction **SI** en guise de test est elle aussi pertinente.

- De **A1 à A6**, Saisissez tantôt des informations et tantôt laissez la cellule vide,
- Sélectionnez les cellules de **B1 à B6**,
- Puis, saisissez la formule **=ESTVIDE(A1)**,
- Validez par **CTRL + Entrée** pour reproduire le calcul.




	A	B
1	18	FAUX
2	59	FAUX
3		VRAI
4	360	FAUX
5	x220	FAUX
6	1220	FAUX

Le résultat est illustré par la figure.

LES FONCTIONS GAUCHE ET DROITE

Ces deux fonctions manipulent des *chaînes de caractères*. Elles fonctionnent de la même manière. Elles permettent d'extraire une partie d'une chaîne soit en partant de la *droite*, soit en partant de la *gauche*. Elles demandent deux paramètres, la cellule contenant le texte et la longueur sur laquelle le texte doit être extrait.

Dans l'exemple illustré ci-dessus, en **B1** est écrit **=GAUCHE(A1;3)**, car seulement les trois premiers caractères en partant de la gauche sont extraits (*bon*). En **B2** est écrit **=DROITE(A2;4)**, car seulement les quatre derniers caractères en partant de la droite sont extraits (*jour*).



	A	B
1	Bonjour	Bon
2	Bonjour	jour
3		



Ces documents **DreamLive** sont libres de droits et ouverts à tous alors, profitez, partagez et appréciez ! Pour nous retrouver sur le Web : <http://www.dreamlive.fr>.

MC  **URS.com** (Zone Cours)
www.mcours.com : Site N° 1 des Cours et Exercices